code:
;-------------------------------------------------
;みんな大好き胖次指令。合意が無いと胖次を見せて要
;裙穿いてる場合には捲ったり、中を覗いたり。
;Ｔ字褲ならお尻を揉む。超低腰なら恥丘をなぞる。高開衩なら鼠径部をなぞる。
;開放型なら隠せてない阴部をじっくり観察しちゃいます
;-------------------------------------------------
@MESSAGE_DAILY_SELECT4
STR:日常イベント名 = %STR:選択肢%

SELECTCASE STR:日常イベント名
CASE ""
	STR:日常イベント名 = %DAILYCOMNAME(4)%
CASE "掀起裙子"
	;相手に隙がある時には你が掀起裙子確率が高い(隙を突く感じ)
	IF CFLAG:睡眠 || CFLAG:刚睡醒 || (COND("打招呼前") && RAND:3) || RAND:(1 + CFLAG:隙有)
	;相手が主动掀起裙子(他に人が居たら条件を厳しくするか？)
	ELSEIF EQUIP:裙 && (ABL:露出癖 >= 3 || ABL:奉仕精神 >= 5) && CFLAG:喧嘩中 == 0
		STR:日常イベント名 = 撩起裙子
	ENDIF
ENDSELECT

;-------------------------------------------------
;ここは起きてる場合に来る
;裙めくってない場合はまずそれから
;めくり済みなら撫でるor観察
;-------------------------------------------------
@TEXT_SELECT_DAILY4
#DIM LCOUNT

STR:選択肢 =
STR:禁止選択肢 =

;口上側で選択肢を自由に作ることができる
TRYCALLFORM DAILY4_K{NO:TARGET}_ADDSELECT
STR:追加選択肢 = /%STR:選択肢%

;裙履いていて捲ってない＆下が何か不明なら掀起裙子
IF EQUIP:裙 && CFLAG:掀起裙子 == 0 && CHECK_CLO("掀起裙子の価値无") == 0
	STR:選択肢 += "掀起裙子/"
	SIF COND("合意：检查内褲")
		STR:選択肢 += "撫摸股間/"
ELSE
	;胖次の代わりに何かを着ている場合
	IF CHECK_CLO("贴身制服露出") || CHECK_CLO("泳装姿")
		STR:選択肢 += "股間観察/"
	ELSE
		;胖次の上に何かを履いている場合
		IF CHECK_CLO("外套露出")
			IF CHECK_OUTER("布鲁马")
				STR:選択肢 += "检查布鲁马/"
			ELSE
				STR:選択肢 += "股間観察/"
			ENDIF
			;SIF CHECK_OUTER("前開可") && COND("合意：检查内褲")
			;	STR:選択肢 += "打開拉链/"
		ENDIF
		;没穿内褲が明確な場合
		IF CHECK_CLO("股間露出")
			STR:選択肢 += "股間観察/"
		ELSEIF EQUIP:胖次 || COND("胖次詳細確認") == 0
			STR:選択肢 += "检查内褲/"
		ENDIF
	ENDIF

	IF CHECK_CLO("Ｖ露出")
		SIF COND("合意：检查阴部")
			STR:選択肢 += "撫摸股間/"
	ELSEIF COND("合意：检查内褲")
		STR:選択肢 += "撫摸股間/"
	ENDIF

	IF COND("合意：检查阴部") || CFLAG:检查阴部 == 0
		STR:選択肢 += "检查阴部/"
	ENDIF
	SIF COND("检查菊穴出現") && (COND("合意：检查阴部") || CFLAG:检查阴部 == 0)
		STR:選択肢 += "检查菊穴/"
ENDIF

VARSET LOCALS
SPLIT STR:選択肢, "/", LOCALS
STR:選択肢 =
FOR LCOUNT, 0, 10
	SELECTCASE LOCALS:LCOUNT
	;男性では起きないようにしておく
	CASE "撫摸股間"
		SIF TALENT:男性
			CONTINUE
	ENDSELECT

	STR:選択肢 += @"%LOCALS:LCOUNT%/"
NEXT

CALL SELECT_BAN

;-------------------------------------------------
;ここにはセリフのみを置く
;-------------------------------------------------
@MESSAGE_DAILY4
#DIM ACTOR
#DIM TEXT_LEN
SIF TFLAG:地の文スキップ
	RETURN 0

SELECTCASE STR:日常イベント名
CASE "检查菊穴"
	PRINTFORM %CALLNAME:MASTER%
	IF COND("性処理：Ａ挿入直後")
		PRINTFORMW 要求刚刚给自己进行完性処理的%CALLNAME:TARGET%、把尻穴张開好仔细看清楚。
	ELSE
		SIF CHARA_NUM("室内の邪魔者") >= 1
			PRINTFORM 完全不管周围还有人、
		SIF CHECK_CLO("裸体")
			PRINTFORM 全裸的
		PRINTFORMW 要求%CALLNAME:TARGET%把尻穴张開。
	ENDIF
CASE "检查阴部", "股間格子"
	IF CFLAG:睡眠
		;連続実行
		IF CONDS("直前行動") == STR:日常イベント名
			PRINTFORMW %CALLNAME:MASTER%把%CALLNAME:TARGET%掰的更開、继续仔细地観察自己想看的东西。
		;中出し直後
		ELSEIF INSERT("挿入")
			PRINTFORMW %CALLNAME:MASTER%中断了动着腰侵犯%CALLNAME:TARGET%的动作、打算確認一下刚刚使用完的阴部的状况。
		;已中出み(ひとまず下のものと一緒にする)
		;ELSEIF CHOSE_DEVENT("睡眠姦Ｖ中出し")
		;使用後
		ELSEIF EXP_GAIN("睡眠姦経験")
			PRINTFORMW %CALLNAME:MASTER%的脸凑近了正在睡觉的%CALLNAME:TARGET%的股間、打算確認一下使用过的阴部的状况。
		ELSE
			PRINTFORMW %CALLNAME:MASTER%的脸凑近了正在睡觉的%CALLNAME:TARGET%的股間、仔细地観察了一下。
		ENDIF
	ELSEIF CONDS("直前行動") == STR:日常イベント名 && COND("合意：检查阴部") == 0
		PRINTFORMW %CALLNAME:MASTER%\@ MARK:屈服刻印 >= 3 ? 要求 # 拜托 \@%CALLNAME:TARGET%再一次把大腿张開。
	ELSE
		PRINTFORM %CALLNAME:MASTER%
		IF COND("性処理直後")
			PRINTFORMW \@ MARK:屈服刻印 >= 3 ? 要求 # 拜托 \@刚刚给自己进行完性処理的%CALLNAME:TARGET%、把大腿张開好仔细看清楚。
		ELSE
			SIF CHARA_NUM("室内の邪魔者") >= 1
				PRINTFORM 完全不管周围还有人、
			IF CHECK_CLO("裸体")
				PRINTFORM 全裸的
			ELSEIF EQUIP:胖次 == 0 && CONDS("直前行動") == "胖次交易"
				PRINTFORM 交易之后理所当然是没有胖次的
			ENDIF
			PRINTFORM %CALLNAME:TARGET%將
			SELECTCASE CONDS("直前行動")
			CASE STR:日常イベント名
				PRINTFORM 再一次张開双腿的
			CASE "给对方剃毛"
				PRINTFORM \@ CHARA_NUM("室内の邪魔者") >= 1 ? # 、 \@将腰挺出来露出了刚刚被剃得干干净净的地方
			CASEELSE
				PRINTFORM 打開自己的双腿
			ENDSELECT
			PRINTFORMW \@ MARK:屈服刻印 >= 3 ? 要求 # 请求 \@了。
		ENDIF
	ENDIF
CASE "脱掉内裤"
	IF CHECK_PANTIES("前貼")
		PRINTFORMW %CALLNAME:MASTER%は%CALLNAME:TARGET%の股間に張り付けられている前貼を剥がし、%TEXTS("大陰唇：視覚")%が良く見えるようにした。
	ELSEIF CHECK_PANTIES("Ｃ胖次")
		PRINTFORMW %CALLNAME:MASTER%は%CALLNAME:TARGET%の股間に装着されているＣ胖次を外し、%TEXTS("大陰唇：視覚")%が良く見えるようにした。
	ELSE
		PRINTFORMW %CALLNAME:MASTER%趁着%CALLNAME:TARGET%正在睡觉、把她的%PANTIES(TARGET, "胖次")%脱掉以便进一步観察。
	ENDIF
;检查内褲
CASE "检查内褲"
	IF COND("胖次詳細確認")
		CALL PRINT_CUTIN, "胖次正面", TARGET
		IF CHECK_PANTIES("前貼")
			PRINTFORMW %CALLNAME:MASTER%は%CALLNAME:TARGET%の股間に顔を埋めて調査を続行した。
		ELSE
			PRINTFORMW %CALLNAME:MASTER%将脸贴近%CALLNAME:TARGET%的股間、继続进行%PANTIES(TARGET, "胖次")%的調査。
		ENDIF
	ELSE
		PRINTFORM %CALLNAME:MASTER%
		IF CHECK_CLO("外套露出") && CHECK_OUTER("连褲襪") == 0
			PRINTFORM %CALLNAME:TARGET%将%OUTER(TARGET, "略")%脱下、
			IF COND("胖次形状確認")
				IF EQUIP:胖次 == 0
					PRINTFORM 把什么都没穿的闷热的股间
				ELSE
					PRINTFORM 把穿着%PANTIES(TARGET, "種類")%的股間
				ENDIF
			ELSE
				PRINTFORM 把胖次穿着的状态
			ENDIF
		ELSEIF CHECK_PANTIES("前貼")
			PRINTFORM %CALLNAME:TARGET%的股間に顔を近づけて
		ELSE
			;裙めくった？
			SIF CONDS("直前行動") == "掀起裙子"
				PRINTFORM 不满足于仅仅只是翻開裙子、
			SIF CHECK_CLO("连褲襪露出")
				PRINTFORM 隔着连裤袜
			PRINTFORM 向着%CALLNAME:TARGET%的\@ EQUIP:胖次 ? %PANTIES(TARGET, "胖次")% # 什么都没有穿的股间 \@凑近了脸
		ENDIF
		PRINTFORMW 仔细地観察了一番。
	ENDIF
CASE "检查布鲁马"
	CALL PRINT_CUTIN, "見たまま正面", TARGET
	IF CFLAG:睡眠
		PRINTFORMW %CALLNAME:MASTER%把正在睡觉的%CALLNAME:TARGET%的双脚打開、打算比平常更仔细地鉴赏一下布鲁马。
	ELSE
		PRINTFORMW %CALLNAME:MASTER%%TEXTR("斜着眼観察/偷偷地偷窥/不经意间移动視線看着")%%CALLNAME:TARGET%穿着的布鲁马。
	ENDIF
CASE "打開拉链"
	CSTR:外套 += "前開/"
	SIF CHECK_CLO("CIP発生")
		CALL PRINT_CUTIN, "見たまま正面", TARGET
	PRINTFORMW %CALLNAME:MASTER%将%CALLNAME:TARGET%穿着的%OUTER(TARGET, "略")%的纽扣打開、%TEXTR("拉下了拉链/拉下了拉链让前面的部分完全露了出来/把拉链全部拉了下来向里面窥视着")%。

CASE "股間観察"
	IF CFLAG:睡眠
		CALL PRINT_CUTIN, "見たまま正面", TARGET
		PRINTFORMW %CALLNAME:MASTER%把正在睡觉的%CALLNAME:TARGET%的双脚打開、把脸凑近了股間仔细地観察了一番。
	;各時間帯の初回
	ELSEIF CFLAG:股間確認済 == 0
		CALL PRINT_CUTIN, "見たまま正面", TARGET
		PRINTFORM %CALLNAME:MASTER%把%CALLNAME:TARGET%的
		IF CHECK_CLO("勒紧的")
			IF CHECK_CLO("外套露出")
				PRINTFORM %OUTER(TARGET, "略")%
			;胖次の代わりに何かを着ている場合
			ELSEIF CHECK_CLO("贴身制服露出")
				PRINTFORM %BODYS(TARGET, "略")%
			ELSEIF CHECK_CLO("泳装姿")
				IF CHECK_PANTIES("比基尼")
					PRINTFORM %PANTIES(TARGET, "略")%
				ELSE
					PRINTFORM 泳装
				ENDIF
			;これは多分ないはずだが一応
			ELSE
				PRINTFORM 股布
			ENDIF
			PRINTFORM 勒紧的\@ COND("陰毛") <= 3 ? 無毛的 # \@股間
		ELSE
			PRINTFORM 从腰部到股間
		ENDIF
		PRINTFORMW %TEXTR("斜着眼観察/偷偷地偷窥/无意识地扫视")%。
	ELSEIF CHECK_CLO("勒紧的") && TALENT:女性
		CALL PRINT_CUTIN, "見たまま正面", TARGET
		PRINTFORMW %CALLNAME:MASTER%把脸凑近%CALLNAME:TARGET%的股間、仔细观察%TEXTR("恥丘/恥骨周辺")%的盛况和陰唇被勒紧的情况。
	ELSE
		PRINTFORMW %CALLNAME:MASTER%将脸靠近%CALLNAME:TARGET%の\@ CHECK_CLO("股間露出") ? 什么都没有穿的 # \@股間详细地调查了一番。
	ENDIF
CASE "撫摸股間"
	SIF CHECK_CLO("CIP発生")
		CALL PRINT_CUTIN, "見たまま正面", TARGET
	PRINTFORM %CALLNAME:MASTER%
	SELECTCASE CONDS("直前行動")
	CASE STR:日常イベント名
		IF PENIS(TARGET) == 0 && CHECK_OUTER("前開") + CHECK_OUTER("迷你")
			IF CHECK_OUTER("迷你")
				PRINTFORM %CALLNAME:TARGET%的小小的%OUTER(TARGET, "略")%
			ELSE
				PRINTFORM %CALLNAME:TARGET%的%OUTER(TARGET, "略")%
			ENDIF
			PRINTFORM 中%TEXTR("伸出/放入/插入")%手、
			IF EQUIP:胖次 == 0 || CHECK_PANTIES("Ｖ露出")
				PRINTFORMW 向股间的%TEXTR("纵裂/裂缝")%插入手指%TEXTR("持續地揉弄著/来回抽插着/持续地玩弄着陰核")%。
			ELSE
				PRINTFORMW %PANTIES(TARGET, "胖次")%越しに股間の%TEXTR("纵裂/割れ目")%%TEXTR("を執拗に弄り続け/の柔らかさを味わっ/を指先でなぞり続け")%た。
			ENDIF
		ELSE
			PRINTFORMW %TEXTR("持續地揉弄著/再度將手伸向/伸出手繼續揉弄")%%CALLNAME:TARGET%的股間。
		ENDIF
		RETURN 1
	;""の上はやや特殊分岐
	CASE "掀起裙子", "撩起裙子"
		PRINTFORM 向着%TEXTR(@"被看得清清楚楚的/暴露在眼前的/隐藏在%CLO("掀起裙子")%下的")%
		PRINTFORM %CALLNAME:TARGET%的%TEXTR("大腿的间隙/下腹部/大腿根部")%伸出了手
		IF CHECK_CLO("Ｖ露出")
			PRINTFORMW %TEXTR("用手捂住露出来的阴部抚弄/将手指插入露出来的裂隙中抚摸敏感的黏膜/用手指玩弄露出来的大阴唇")%。
		ELSE
			PRINTFORMW 钻入%PANTIES(TARGET, "略")%中%TEXTR("抚摸阴部/将手指插入裂隙中搅动/享受着大阴唇的触感")%。
		ENDIF
		RETURN 1
	CASE "把手放在大腿上"
		PRINTFORM 放在%CALLNAME:TARGET%的大腿上的兽不安分地滑向%TEXTR("大腿内側/下腹部/股间")%
		IF CHECK_CLO("Ｖ露出")
			PRINTFORMW 、抚摸完全裸露出来的股间。
		ELSEIF CHECK_CLO("连褲襪露出")
			PRINTFORMW 、穿过%CLO("股間")%享受着%TEXTR("用手心包裹住阴部的触感/用指尖玩弄的触感")%。
		ELSEIF CLO("股間") != ""
			PRINTFORMW 、%TEXTR(@"钻入%CLO("股間")%中抚摸股间/插入%CLO("股間")%中用手指玩弄蜜裂/手指插入%CLO("股間")%中享受大阴唇的触感")%。
		ELSE
			PRINTFORMW 撫摸股間。
		ENDIF
		RETURN 1
	CASE "给对方剃毛"
		PRINTFORM 对着%TEXTR("刚刚剃毛的/剃毛结束之后的/刚刚变成光滑的样子不久的")%%CALLNAME:TARGET%的下腹部%TEXTR("伸出了手指/用手掌抚摸了起来/伸出了手")%、
		PRINTFORMW 享受着%TEXTR("没有任何毛发残留/漂亮地剃干净/顺滑")%的触感。
		RETURN 1
	CASE ""
		IF CFLAG:打招呼された == 0
			PRINTFORM 明明还没有打过招呼呢
		ELSE
			PRINTFORM 慢慢地
		ENDIF
	CASE "叫醒"
		PRINTFORM %TEXTR("剛剛睡醒頭還昏昏沉沉的/正在伸懶腰的/似乎還沒睡醒眼神朦朧的")%
	CASE "打招呼", "打招呼ハグ", "打招呼接吻"
		PRINTFORM 互相打著招呼的同時
	CASE "甜言蜜语"
		PRINTFORM 小声地奉承着的同時、一边对着%TEXTR("自然地動作着的/慢吞吞的/感觉也不是那么坏的/害羞着的")%的对方、
	CASE "夸奖"
		PRINTFORM 一边说着赞美的话、一边对着%TEXTR("自然地動作着的/慢吞吞的/被突入了破绽的/感觉也不是那么坏的/脸頬放松了下来的")%的对方、
	CASE "Talk", "Ｈ的話"
		PRINTFORM 一边%TEXTR("天南海北地闲聊着/滔滔不绝地聊着/随随便便地聊着")%、一边
	CASE "膝枕"
		PRINTFORM %TEXTR("頭抬了一下/咕噜地滚了一下/迅速改変了体勢")%、对着正给你膝枕的%CALLNAME:TARGET%、
	CASE "露脐装超短裙胖次チラ"
		PRINTFORM 刚掀起了胖次就
	CASEELSE
		PRINTFORM %TEXTS("行動継続中の")%
	ENDSELECT

	IF EQUIP:裙
		PRINTFORM 将手%TEXTR("伸入/插入")%%CALLNAME:TARGET%的%CLO("掀起裙子：詳細")%中、
		IF CHECK_CLO("连褲襪露出")
			PRINTFORMW 钻入%CLO("股間")%中%TEXTR("用手包裹住/用手指挑逗着")%享受着股间的触感。
		ELSE
			PRINTFORMW %TEXTR("用手心包裹住股间抚弄着/用手指挑逗着股间的蜜裂/将手放入大腿之间")%。
		ENDIF
		RETURN 1
	ELSE
		PRINTFORM 向着%CALLNAME:TARGET%的%TEXTR("大腿之间/下腹部/大腿根")%中%TEXTR("放入手/伸出手")%
		IF CHECK_CLO("Ｖ露出")
			PRINTFORMW 抚摸露出来的股间。
		ELSEIF CHECK_CLO("连褲襪露出")
			PRINTFORMW 钻入%CLO("股間")%中%TEXTR("用手包裹住/用手指挑逗着")%享受着股间的触感。
		ELSEIF CLO("股間") != ""
			PRINTFORMW 、%TEXTR(@"钻入%CLO("股間")%中抚摸股间/插入%CLO("股間")%中用手指玩弄蜜裂/手指插入%CLO("股間")%中享受大阴唇的触感")%。
		ELSE
			PRINTFORMW 撫摸股間。
		ENDIF
	ENDIF
CASE "掀起裙子"
	PRINTFORM %CALLNAME:MASTER%
	SELECTCASE CONDS("直前行動")
	CASE "叫醒"
		PRINTFORM 掀起了%TEXTR("刚起床、頭脑似乎还没有清醒过来/伸着懒腰打着哈欠/睡意还没有完全消除")%、还睁着惺忪的睡眼的%CALLNAME:TARGET%的
	CASE "打招呼"
		PRINTFORM 一边和%CALLNAME:TARGET%打着招呼、一边掀起了
	CASE "甜言蜜语"
		PRINTFORM 小声地奉承着的同時、一边掀起了%TEXTR("自然地動作着的/慢吞吞的/感觉也不是那么坏的/害羞着的")%%CALLNAME:TARGET%的
	CASE "夸奖"
		PRINTFORM 一边说着赞美的话、一边掀起了%TEXTR("自然地動作着的/慢吞吞的/被突入了破绽的/感觉也不是那么坏的/脸頬放松了下来的")%%CALLNAME:TARGET%的
	CASE "Talk", "Ｈ的話"
		PRINTFORM 一边和%CALLNAME:TARGET%%TEXTR("天南海北地闲聊着/滔滔不绝地聊着/随随便便地聊着")%、一边掀起了
	CASE "膝枕"
		PRINTFORM %TEXTR("頭抬了一下/咕噜地滚了一下/迅速改変了体勢")%、掀起了正给你膝枕的%CALLNAME:TARGET%的
	CASE "露脐装超短裙胖次チラ"
		PRINTFORM 掀起了刚刚偷窥过的%CALLNAME:TARGET%的
	CASEELSE
		;相手が寝ている
		IF CFLAG:睡眠
			PRINTFORM 掀起了無防備地睡着的%CALLNAME:TARGET%的
		;打招呼してない
		ELSEIF COND("打招呼前")
			PRINTFORM 代替招呼、掀起了%CALLNAME:TARGET%的
		ELSE
			PRINTFORM %CALLNAME:TARGET%的
		ENDIF
	ENDSELECT
	PRINTFORMW %CLO("掀起裙子：詳細")%。
CASE "撩起裙子"
	IF CFLAG:躁动难忍 + CFLAG:甜言蜜语 + CFLAG:夸奖
		CALL EMOTION_PLUS
	ELSEIF TALENT:害羞
		CALL EMOTION_PLUS, "害羞"
	ELSE
		CALL EMOTION_PLUS, "小悪魔"
	ENDIF
	PRINTFORM %TEXTR("察觉到了你的欲求/顺应着你的態度/感觉到了你的視線")%的
	PRINTFORM %CALLNAME:TARGET%把%CLO("掀起裙子")%
	IF TALENT:強気 + TALENT:傲慢 && RANDIF(1, TALENT:粗野, TALENT:高贵, TALENT:開朗, TALENT:内気)
		PRINTFORM %TEXTR("用挑衅的动作/一边轻轻晃动着腰肢/自信满满地/一边挺着胸")%
	ELSEIF TALENT:内気 && RANDIF(1, TALENT:粗野, TALENT:高贵, TALENT:開朗)
		PRINTFORM %TEXTR("提心吊胆地/沉默地/战战兢兢地")%
	ELSEIF TALENT:開朗 && RANDIF(1, TALENT:粗野, TALENT:高贵)
		PRINTFORM %TEXTR("能看到肚脐程度地/看起来很開心地/兴奋地")%
	ELSEIF TALENT:高贵 && RANDIF(1, TALENT:粗野)
		PRINTFORM %TEXTR("优雅地/充满感觉地/瀟洒地")%
	ELSEIF TALENT:粗野
		PRINTFORM %TEXTR("啪地一口气地/一口气地/能看到肚脐程度地")%
	ELSE
		PRINTFORM %TEXTR("自発的/慢慢地/在你提出来之前")%
	ENDIF
	PRINTFORM 用手慢慢向上卷起、将裙下的秘密之处
	IF ABL:奉仕精神 > ABL:露出癖
		PRINTFORMW 展示了出来。
	ELSE
		PRINTFORMW 展示了出来。
	ENDIF
ENDSELECT

;-------------------------------------------------
;主に胖次の描写などをする。STR:日常イベント名 は決定後。
;-------------------------------------------------
@AFTERMESSAGE_DAILY4
;起きうるイベントの判定に使う
#DIMS EVENT_TABLE
#DIMS TEXT_DETAIL
;阴茎ルートor阴部ルート
#DIM IS_PENISROUTE
IS_PENISROUTE = 0

SELECTCASE STR:日常イベント名
CASE "检查阴部", "检查菊穴", "股間格子"
	IF CFLAG:睡眠 == 0 && COND("合意：检查阴部") == 0
		IF CHOSE_DEVENT(STR:日常イベント名)
			PRINTFORMW 這次也被%CALLNAME:TARGET%討厭了。
		ELSE
			PRINTFORML 被%CALLNAME:TARGET%討厭了。
		ENDIF
		CALL PRINT_STRW, "實行這種行為需要先取得_黄色_[检查阴部]合意_。"
		RETURN 0
	ENDIF

	IF CFLAG:睡眠
		;連続実行
		IF CONDS("直前行動") == STR:日常イベント名
			PRINTFORM %CALLNAME:TARGET%不知道自己最珍贵的地方已经被%TEXTS("間柄の", MASTER)%%CALLNAME:MASTER%一个角落都不剩得看了个精光、
			PRINTFORMW %TEXTR("露出了毫無防備的睡顔/依然睡的很舒服的様子/嘶嘶得静静呼吸着")%……
		;中出し直後
		ELSEIF INSERT("挿入")
			IF TALENT:淫壷 && RAND:3 == 0
				PRINTFORMW 被%CALLNAME:MASTER%尽情享用过的%CONDS("間柄")%的膣壁帯着熱气、刚被注入进去的濃稠的精液就像流出来的口水一様垂落在阴道口、随着呼吸妖艳地蠕動着…
			ELSEIF RAND:2 == 0
				PRINTFORMW 被满满地中出的小穴、因为快感而张開着阴道口、沾满了精液的粘膜清晰可见…
			ELSE
				PRINTFORMW 刚刚被中出过的%CALLNAME:TARGET%的膣口就像喘气一様不断開合着、调皮的精液从里面断断续续地垂落下来、散发着濃厚的性臭味…
			ENDIF
		;已中出み(ひとまず下のものと一緒にする)
		;ELSEIF CHOSE_DEVENT("睡眠姦Ｖ中出し")
		;使用後
		ELSEIF EXP_GAIN("睡眠姦経験")
			IF ABL:Ｖ拡張 >= 3 && RAND:3 == 0
				PRINTFORM 从被大大扩张開来的%CALLNAME:TARGET%的膣穴之中、被撹拌到发泡的愛液
				SIF CHOSE_DEVENT("睡眠姦Ｖ中出し")
					PRINT 裹挟着被注入体内的灼热白浊一起、
				PRINTFORMW 黏黏糊糊地垂落下来…
			ELSEIF ABL:Ｖ拡張 >= 2 && RAND:2 == 0
				PRINTFORMW 被%CALLNAME:MASTER%刚刚侵犯过而不能閉合的肉洞窟的深处、渴求着更多的快感的膣的粘膜还在蠢蠢欲动…
			ELSEIF RAND:3 == 0 && SIZE("阴茎", PLAYER) >= 2 && AGE(TARGET) == 1
				PRINTFORMW %CALLNAME:TARGET%带着稚气的身体因为被插入了过大的阴茎、膣口大大的张開着、甚至能够直接看到最里面的内壁…
			ELSEIF RAND:2
				PRINTFORMW 刚刚被侵犯过的那里因为流出来的发泡的愛液而变得黏黏糊糊的、散发着非常厉害的%TEXTS("气味")%…
			ELSE
				PRINTFORMW 被阴茎扩张開的%CALLNAME:TARGET%的膣口、因为快感而张開着阴道口、内部柔软粘膜一览无余…
			ENDIF
		ELSE
			IF TALENT:肤色 && TALENT:男性 == 0 && RAND:3 == 0
				SELECTCASE CONDS("肤色")
				CASE "褐色肌"
					PRINTFORM 褐色的肌肤
				CASE "色白"
					PRINTFORM 通透的肌肤
				CASE "青肌"
					PRINTFORM 青色的肌肤
				CASEELSE
					PRINTFORM 周辺的肌肤
				ENDSELECT
				PRINTFORM 因为%TEXTR("対比度/颜色対比/色彩差异")%而%TEXTR("让人移不開眼睛的/美丽的/鮮艳的")%粘膜
			ELSE
				PRINTFORM 以大腿张開的体勢
				IF TALENT:男性 || (TALENT:扶她 && RAND:2)
					PRINTFORM 無防備地暴露出%CONDS("間柄")%的
					IF TALENT:陰蒂環  && RAND:2
						PRINTFORM 將被反射著光芒的穿環穿刺著的
					ELSE
						PRINTFORM %TEXTS("阴茎：視覚", TARGET)%
					ENDIF
					PRINTFORM 阴茎は
					IS_PENISROUTE = 1
				ELSE
					PRINTFORM %TEXTR("無防備地暴露出/毫不憐惜地張開着/在眼前開花的")%%CONDS("間柄")%的
					IF TALENT:陰蒂環 + TALENT:陰唇環 && RAND:2
						PRINTFORM 穿環\@ (TALENT:陰蒂環 && TALENT:陰唇環) ? 扎满的 # 閃爍的 \@
					ELSEIF COND("陰毛") == 7 && RAND:2
						PRINTFORM 毛茸茸的
					ELSEIF (COND("陰毛") == 1 || COND("陰毛") == 3) && RAND:2
						PRINTFORM %TEXTR("無毛/滑滑")%的
					ELSEIF COND("陰毛") == 2 && RAND:2
						PRINTFORM 漂亮脱毛过的
					ELSEIF AGE(TARGET) >= 4
						PRINTFORM 成熟的
					ELSEIF AGE(TARGET) <= 1
						PRINTFORM 年幼
					ENDIF
					PRINTFORM 阴部
				ENDIF
			ENDIF
			;体育後
			IF CFLAG:体育 && COND("昼")
				PRINTFORM %TEXTR("身体运動过後的/刚刚運動完的/体育直後的")%原因、
				IF IS_PENISROUTE
					PRINTFORM %TEXTR(@"散发着微微臭味/蒸发開来的强烈的%TEXTS("气味")%飘散着/混杂着汗液和尿液的污垢轻轻地附着")%、
				ELSE
					PRINTFORM %TEXTR(@"散发着忍受程度以内的臭味/蒸发開来的强烈的%TEXTS("气味")%飘散着/混杂着汗液和尿液的污垢轻轻地附着/以汗为中心的分泌物的强烈的%TEXTS("气味")%飘散着")%、
				ENDIF
			;水泳後
			ELSEIF CFLAG:水泳 && COND("昼")
				PRINTFORMW 水泳之後没做什么事所以几乎無臭、即使用舌頭舔下去也没有不好的味道、总感觉少了些什么……
				RETURN 1
			ELSE
				PRINTFORM %TEXTR(@"\@ COND("昼") ? 午前中 # 一天里 \@蓄積下来隐约的性臭飘散着/打扮得很漂亮但还是散发着微弱臭味/说不上很好也不会很糟的%TEXTS("气味")%散布開来")%、
			ENDIF
			IF TALENT:陰蒂環 + TALENT:陰唇環 && RAND:3 == 0
				PRINTFORM %TEXTR("享乐般玩弄着穿環/玩耍一样拉着穿環/重新観察穿環")%
			ELSEIF TALENT:男性 == 0 && TALENT:処女 == 0 && RAND:4 == 0
				PRINTFORM 把手指伸进阴道看着对方反应
			ELSEIF IS_PENISROUTE
				PRINTFORM %TEXTR("不顾旁人眼光完全入迷/咕滋咕滋地玩弄/用手指把皮抻起玩弄/在近距离观察")%
			ELSE
				PRINTFORM %TEXTR("不顾旁人眼光完全入迷/咕滋咕滋地扩開玩弄/咕滋咕滋地玩弄/用手指把陰唇抻起玩弄/在近距离观察")%
			ENDIF
			PRINTFORMW 的%CALLNAME:MASTER%通过鼻孔感受到了对方下半身强烈的诉求……
		ENDIF
	ELSEIF COND("合意：检查阴部") || EQUIP:Ｈ的奖励
		IF CHECK_CLO("连褲襪露出") && CHECK_OUTER("開洞")
			PRINTFORM 唰地把%OUTER(TARGET, "略")%脱下
		ELSEIF EQUIP:外套 && CHECK_OUTER("めくり可能") == 0 && CHECK_OUTER("開洞")
			PRINTFORM 嗖嗖地把%OUTER(TARGET, "略")%脱下
		ELSEIF STR:日常イベント名 == "检查菊穴"

		ELSE
			PRINTFORM 坐在椅子上的
		ENDIF
		PRINTFORM %CALLNAME:TARGET%

		IF TALENT:害羞 || EQUIP:Ｈ的奖励
			PRINTFORM %TEXTR("有点害臊的/看起来有些犹豫地/羞恥的身体像火一样红的")%
		ELSEIF TALENT:不知羞恥 && RAND:3 == 0
			PRINTFORM %TEXTR("一点都不觉得羞耻地/脸色一丝不变地/若无其事地")%
		ELSEIF TALENT:開朗
			PRINTFORM %TEXTR("轻松愉快地/笑着/姿势不雅地/挺起腰来")%
		ELSEIF TALENT:強気 && RANDIF(1, TALENT:高慢)
			PRINTFORM %TEXTR("主动大幅度/不避開视线地")%
		ELSEIF TALENT:坦率
			PRINTFORM %TEXTR("听从了指令/従順地/怯生生地")%
		ELSEIF TALENT:高慢
			PRINTFORM %TEXTR("自信满满地/满怀優越感地/慢慢地")%
		ELSE
			PRINTFORM %TEXTR("按照你期望地/従順地/回应要求地")%
		ENDIF

		SELECTCASE STR:日常イベント名
		;肛門ルート
		CASE "检查菊穴"
			SIF CHECK_PANTIES("Ａ露出") && STR:日常イベント名 == "检查菊穴" && PANTIES(TARGET, "胖次") == "胖次"
				PRINTFORM 穿着%PANTIES(TARGET)%
			PRINTFORM 伸出屁股、
		CASEELSE
			PRINTFORM 張開腿、
		ENDSELECT

		IF CHECK_PANTIES("Ａ露出") && STR:日常イベント名 == "检查菊穴"

		ELSEIF CHECK_PANTIES("Ｖ露出") && STR:日常イベント名 != "检查菊穴"

		ELSEIF CHECK_PANTIES("前貼")
			PRINTFORM 撕下了貼紙
		ELSEIF CHECK_PANTIES("Ｃ胖次")
			PRINTFORM 脫下了Ｃ胖次
		ELSEIF CHECK_CLO("连褲襪露出")
			SIF EQUIP:胖次
				PRINTFORM 错開%PANTIES(TARGET, "略")%
		ELSEIF EQUIP:外套 && CHECK_OUTER("めくり可能") == 0
			SIF EQUIP:胖次
				PRINTFORM 错開%PANTIES(TARGET, "略")%
		ELSEIF CHECK_CLO("外套露出") || CHECK_CLO("贴身制服露出") || CHECK_CLO("胖次露出")
			IF CHECK_OUTER("带拉链的") || CHECK_BODYS("带拉链的") || CHECK_PANTIES("带拉链的")
				PRINTFORM 胯股之间的带拉链打開
			ELSE
				PRINTFORM 错開%CLO("股間")%
			ENDIF
		ENDIF
		;使用直後
		IF COND("性処理：生挿入直後") || COND("性処理：挿入直後")
			PRINTFORM %CALLNAME:MASTER%对着使用之後
			IF COND("性処理：生挿入直後")
				PRINTFORM 精液滴落的
			ELSEIF STR:日常イベント名 == "检查菊穴"
				PRINTFORM %TEXTR("松动的/一张一合的/甜蜜地閉合的")%
			ELSE
				PRINTFORM 的
			ENDIF
			SELECTCASE STR:日常イベント名
			;肛門ルート
			CASE "检查菊穴"
				IF COND("性処理：生挿入直後") && (RAND:2 || TALENT:縦深菊穴 == 0)
					PRINTFORM %TEXTR("松动的/甜蜜地閉合的")%
				ELSEIF TALENT:縦深菊穴
					PRINTFORM 縦向褶皱的
				ENDIF
				PRINTFORM 肛門
			CASEELSE
				PRINTFORM 股间的穴
			ENDSELECT
			PRINTFORMW 用手指压下去打開仔细观察……
		ELSEIF STR:日常イベント名 == "检查菊穴"
			IF COND("身嗜み：Ａ拡張")
				PRINTFORMW 对着%CALLNAME:MASTER%展示%TEXTS("肛門周辺")%……
			ELSE
				PRINTFORMW 朝向%CALLNAME:MASTER%展示%TEXTS("肛門周辺")%……
			ENDIF
		;干劲性処理直後
		ELSEIF CONDS("直前行動") == "拜托処理性欲" && CFLAG:躁动难忍 + COND("発情期")
			IF TALENT:男性
				PRINTFORMW 面向%CALLNAME:MASTER%露出了因奉仕行為而完全变硬的%TEXTS("阴茎：尺寸表現付", TARGET)%……
			ELSEIF TALENT:扶她
				PRINTFORMW 面向%CALLNAME:MASTER%露出了因奉仕行為而高昂的秘密場所……
				;PRINTFORM %TEXTR("大概是因为奉仕多少有些兴奋了/可能是进行奉仕行為之後的缘故吧/尽管只是做着奉仕")%、
				;PRINTFORMW 膣口里%TEXTR("漏出了一点透明的潤滑液/的愛液已经快要溢出了/羞耻的粘液已经溢出粘到了肌肤上")%。
			ELSE
				PRINTFORMW 面向%CALLNAME:MASTER%露出了自己秘密的場所……
			ENDIF
		ELSEIF STR:日常イベント名 == "检查阴部" && CHOSE_DEVENT(STR:日常イベント名) == 0
			VARSET LOCALS
			;穿環ルート
			IF TALENT:陰蒂環 && TALENT:陰唇環
				LOCALS = 被%TEXTR("装飾着/付上/镶入")%%TEXTR("复数穿/陰唇和阴蒂")%環の
				IF AGE(TARGET) == 4
					LOCALS += "大人の"
				ELSEIF COND("陰毛") >= 1 && COND("陰毛") <= 3
					LOCALS += "無毛の"
				ENDIF
			ENDIF
			;外性器評価ルート
			IF COND("Ｖ熟達")
				SELECTCASE CSTR:学年
				CASE "初等部", "中等部", "１年生"
					LOCALS:1 = %TEXTR(@"%CSTR:学年%の癖的\/还仍是%CSTR:学年%就已经\/与兒童的外表相比难以想象地")%%TEXTR("開发济/経験豊富")%の
				CASE "２年生"
					LOCALS:1 = %TEXTR("学園生の癖/２年生の割/同学年とは比較にならない程")%に%TEXTR("使い込まれた/経験豊富な")%
				CASE "３年生"
					LOCALS:1 = %TEXTR("上級生にふさわしい/同学年とは比較にならない程に")%使い込まれた
				CASEELSE
					SIF AGE(TARGET) == 4
						LOCALS:1 = 供%CALLNAME:MASTER%使用的成熟の
				ENDSELECT
			ENDIF
			;陰毛ルート
			SELECTCASE CONDS("陰毛")
			CASE "保养完毕"
				IF COND("自己剃った")
					;自主的にこの状態にしている
					SIF CSTR:陰毛的喜好 == CONDS("陰毛")
						LOCALS:2 = %TEXTR("毎日自己/自主/自発")%%TEXTR("剃得漂亮/将陰毛剃整齐/进行打理/修整陰毛形態/保养陰毛/剃到準備完毕")%的
				ENDIF
			CASE "光滑无毛"
				IF COND("自己剃った")
					;自主的にこの状態にしている
					IF CSTR:陰毛的喜好 == CONDS("陰毛")
						LOCALS:2 = %TEXTR("毎日自己/自主/自発")%%TEXTR("剃毛/修理/修整")%而%TEXTR("十分光滑/無毛")%の
					;你にお願いされてこの状態にしている
					ELSEIF CONDS("身嗜み：陰毛") == CONDS("陰毛")
						LOCALS:2 = 按照要求%TEXTR("剃毛/修理/修整")%て%TEXTR("十分光滑/無毛")%
					ENDIF
				ELSE
					LOCALS:2 = 被剃毛后%TEXTR("光滑/無毛")%の
				ENDIF
			CASE "脱毛処理"
				LOCALS:2 = %TEXTR("没有一根毫毛地脱毛过的/各个角落都完全除毛过的/精心脱毛过的")%%TEXTR("光滑/無毛/光洁")%の
			ENDSELECT
			SIF CONDS("陰毛") != "" && LOCALS:2 == ""
				LOCALS:2 = %TEXTS("陰毛：視覚")%
			LOCALS = %TEXTR(@"%LOCALS%\/%LOCALS:1%\/%LOCALS:2%")%
			IF LOCALS != ""
				PRINTFORMW %LOCALS%%TEXTS("阴部：客観表現")%を曝け出している…
			ELSE
				PRINTFORMW 面向%CALLNAME:MASTER%露出了自己秘密的場所……
			ENDIF
		ELSE
			PRINTFORMW 面向%CALLNAME:MASTER%露出了自己秘密的場所……
		ENDIF
	ENDIF
CASE "检查内褲"
	IF CFLAG:睡眠 == 0 && COND("合意：检查内褲") == 0
		IF CHOSE_DEVENT(STR:日常イベント名)
			PRINTFORMW 這次也被%CALLNAME:TARGET%討厭了。
		ELSE
			PRINTFORML 被%CALLNAME:TARGET%討厭了。
		ENDIF
		CALL PRINT_STRW, "實行這種行為需要先取得_黄色_[检查内褲]合意_。"
		RETURN 0
	ENDIF
	IF COND("胖次形状確認") == 0 && CHECK_CLO("外套露出") && CHECK_OUTER("连褲襪") == 0
		CALL TEXT_PANTU_CHECK, "外套"
	ELSE
		IF EQUIP:胖次
			EVENT_TABLE =
			SIF CHECK_PANTIES("Ｖ露出") && CHOSE_DEVENT("開放型開洞") == 0
				EVENT_TABLE += "開放型開洞/"
			SIF CHECK_PANTIES("高開衩") && CHOSE_DEVENT("高開衩") == 0
				EVENT_TABLE += "高開衩/"
			SIF CHECK_PANTIES("Ｔ字褲") && CHOSE_DEVENT("Ｔ字褲") == 0
				EVENT_TABLE += "Ｔ字褲/"
			SIF CHECK_PANTIES("後開型") && CHOSE_DEVENT("後開型") == 0
				EVENT_TABLE += "後開型/"
			SIF CHECK_PANTIES("Ｃ胖次") && CHOSE_DEVENT("Ｃ胖次") == 0
				EVENT_TABLE += "Ｃ胖次/"

			EVENT_TABLE = %TEXTR(EVENT_TABLE)%

			;何度も出すのもどうかな？
			SIF COND("胖次詳細確認") == 0
				CALL PRINT_CUTIN, "胖次正面", TARGET

			PRINTFORM %CALLNAME:MASTER%
			SELECTCASE EVENT_TABLE
			CASE "開放型開洞"
				PRINTFORMW 仔细地观察同时映入眼帘的陰唇和平常坐下时看不到的開放型胖次的洞部分的结构。
			CASE "後開型"
				PRINTFORM 、对只遮住性器却将肛門露出的构造的%PANTIES(TARGET, "胖次")%
				IF CHECK_CLO("连褲襪露出") && CHECK_OUTER("開洞有连褲襪")
					PRINTFORM 隔着%OUTER(TARGET, "略")%
				ELSE
					PRINTFORM %TEXTR(@"在近距離\/在几乎不能呼吸的距離\/在股間的%TEXTS("气味")%能嗅到的距离")%
				ENDIF
				PRINTFORMW %TEXTR("细心地/充分地/细致地")%欣赏着。
			CASE "高開衩"
				IF RAND:2 && TALENT:男性 == 0
					PRINTFORMW %TEXTR("爱抚/眺望/撫摸")%着从纤细布料的边缘露出来的大陰唇的同时、仔细地欣赏着%CALLNAME:TARGET%的腹股沟。
				ELSE
					PRINTFORMW 在近距離%TEXTR("爱抚/眺望/撫摸")%%CALLNAME:TARGET%的腹股沟的同时、仔细地欣赏着勒进股間的布料。
				ENDIF
			CASE "Ｔ字褲"
				IF RAND:2
					PRINTFORMW 隔着細細的\@ CHECK_PANTIES("丁字褲") ? 丁字褲 # Ｔ字褲 \@纽带、细心的愛抚着%CALLNAME:TARGET%の%TEXTR("肛門/尻穴/排泄孔")%。
				ELSE
					PRINTFORMW 仔细地観察着勒进尻里的\@ CHECK_PANTIES("丁字褲") ? 丁字褲 # Ｔ字褲 \@绳带、和%TEXTR("露出的/藏不住的/色素沉淀的/妖艳的")%肛門褶皱。
				ENDIF
			CASE "Ｃ胖次"
				PRINTFORMW 对着%CALLNAME:TARGET%的股間穿着的Ｃ弦胖次、一边用%TEXTR("手指/手/手掌")%玩弄一边観察着。
			CASEELSE
				IF RAND:2
					PRINTFORM %CALLNAME:TARGET%的%PANTIES(TARGET, "胖次")%を、
					IF TALENT:男性 || (TALENT:扶她 && RAND:2)
						IF CHECK_PANTIES("Ｖ露出")
							PRINTFORM 因为缺少布料而露出的%TEXTR("阴茎/陰嚢")%
						ELSE
							PRINTFORM %TEXTR("隔着布料的阴茎/隔着布料的陰嚢/隔着布料的亀頭")%
						ENDIF
					ELSE
						IF CHECK_PANTIES("Ｖ露出")
							PRINTFORM 因为缺少布料而露出的%TEXTR("膣口/陰核/尿道口")%
						ELSE
							PRINTFORM %TEXTR("隔着布料的膣口/隔着布料的陰核/隔着布料的尿道口/内衣褶皱最多的地方")%
						ENDIF
					ENDIF
					PRINTFORMW 一边用手指玩弄一边观察着。
				ELSE
					PRINTFORM 把脸埋进%CALLNAME:TARGET%的%PANTIES(TARGET, "胖次")%里、
					IF CHECK_PANTIES("Ｖ露出") || RAND:3 == 0
						IF EQUIP:裙 && RAND:2
							PRINTFORM 笼罩在裙子里面
						ELSE
							PRINTFORM 具有生命气息的秘密場所
						ENDIF
						PRINTFORMW 的热腾腾的%TEXTS("气味")%贪婪地享受着。
					ELSEIF RAND:2
						IF RAND:2
							PRINTFORMW 和直接嗅味道不同的、充分地满足了%TEXTS("气味")%嗅觉的癖好。
						ELSE
							SIF CFLAG:汗
								PRINT 濡濕的
							IF TALENT:男性
								PRINTFORMW 股布下男性器的%TEXTS("气味")%尽情地品味着。
							ELSE
								PRINTFORMW 胖次后的阴部%TEXTS("气味")%尽情地品味着。
							ENDIF
						ENDIF
					ELSE
						PRINTFORM 享受隔着%TEXTR("一片布/单薄的布/一片与肌肤接触的布")%和%TEXTR("因興奮浸湿了汗水的/闷热的/微微出汗的")%内股的味道的
						IF TALENT:男性 || (TALENT:扶她 && RAND:2)
							PRINTFORMW %TEXTR("阴茎/陰嚢")%的触感。
						ELSE
							PRINTFORMW 陰唇的柔软。
						ENDIF
					ENDIF
				ENDIF
			ENDSELECT

			;胖次形状ごとの行動した場合はそれを記録
			SIF EVENT_TABLE != ""
				CSTR:DAILYCOM履歴 += @"＠%EVENT_TABLE%\/"
		ENDIF

		;寝てたらここまで
		SIF CFLAG:睡眠
			RETURN 0
		IF ABL:露出癖 >= 5 && RAND:2
			PRINTFORMW %CALLNAME:TARGET%以%TEXTR("迷离的表情/沉溺于欲望的表情/兴奋的样子")%接受%CALLNAME:MASTER%的視姦、漏出了甘甜的吐息……
		ELSEIF ABL:露出癖 >= 3 || (ABL:露出癖 + CFLAG:甜言蜜语 && RAND:2)
			IF EVENT_TABLE == "Ｔ字褲" || (EVENT_TABLE != "高開衩" && RAND:2)
				PRINTFORMW %CALLNAME:TARGET%把手贴在墙上双腿齐肩张開、挺起屁股接受%CALLNAME:MASTER%的視線……
			ELSE
				PRINTFORMW %CALLNAME:TARGET%双腿齐肩张開、挺起腰接收%CALLNAME:MASTER%的視線……
			ENDIF
		ELSEIF TALENT:叛逆
			PRINTFORMW %CALLNAME:TARGET%%TEXTR("有自己想法的样子/没什么干劲的样子/警戒又逞强地")%、轻轻张開双腿一动不动……
		ELSE
			PRINTFORMW %CALLNAME:TARGET%没有抵抗、遵从%CALLNAME:MASTER%的要求轻轻地张開腿一动不动……
		ENDIF
	ENDIF
CASE "脱掉内裤"
	IF CHECK_PANTIES("前貼") || CHECK_PANTIES("Ｃ胖次")
	ELSEIF CFLAG:睡眠 && CHECK_PANTIES("Ｖ露出") == 0
		CALL PRINT_CUTIN, "胖次正面", TARGET
		PRINTFORM %TEXTR("左右抻開/像綾罗糸一様展開/展開")%%PANTIES(TARGET, "種類")%、先不说颜色和形状、
		IF CHECK_PANTIES("開洞带污迹的")
			PRINTFORM %TEXTR("紧贴阴部的/開洞的/最容易脏的")%部分上%TEXTR("染上的微黄的斑点/很浓的污渍/呈条形的脏污")%
		ELSEIF TALENT:男性
			PRINTFORM %TEXTR(@"内側%CALLNAME:TARGET%的污垢/黏上的淡淡的尿污")%
		ELSE
			PRINTFORM %TEXTR(@"内側%CALLNAME:TARGET%的污垢/紧贴%CALLNAME:TARGET%阴部的痕跡/淡淡地沾着条形的污垢")%
		ENDIF
		PRINTFORMW 一览无余地在%CALLNAME:MASTER%眼前暴露出来……
	ENDIF
CASE "检查布鲁马"
	IF CFLAG:睡眠
		PRINTFORM %CALLNAME:TARGET%
		IF RAND:3 == 0
			PRINTFORM 对于布鲁马被捲起、内部笼罩着的%TEXTS("气味")%被嗅到、股布部分的湿気正在被手指触摸確認的事完全不知情、
		ELSEIF RAND:2
			PRINTFORM 对于布鲁马的股間正在被脸摩擦着、%TEXTS("气味")%被嗅到的事完全不知情、
		ELSE
			PRINTFORM 隔着布鲁马的%TEXTS("气味")%被嗅到、\@ (TALENT:男性 || (TALENT:扶她 && RAND:2) )  ? 阴茎 # 阴部 \@被玩弄得無意識扭动着身体的同时、
		ENDIF
		PRINTFORMW 依然沉睡在夢中……
	;はみ胖次してる？
	ELSEIF COND("胖次おぼろげ確認") == 0 && CHECK_CLO("布鲁马はみ胖次可能") && RAND:MAX(5 - TALENT:巨尻 + TALENT:小尻*3, 1) == 0
		PRINTFORM 也许是%TEXTR("没调整好裤口/坐在椅子上时有些偏差/激烈運動所以勒进去")%的原因、
		PRINTFORMW 从%CALLNAME:TARGET%的布鲁马里%TEXTR("能看到一点点胖次的边缘/能看到内衣一闪而过/一点点胖次的边缘露了出来")%……
		PRINTFORMW 虽然只看到一点边缘、但能确定穿着的胖次是%CLOCOLOR(CSTR:胖次, "形容詞")%。
		CALL SETFLAG_DAILY, "胖次色確認"
	ENDIF
CASE "股間観察"
	IF CFLAG:睡眠
		PRINTFORM %CALLNAME:TARGET%
		IF RAND:3 == 0
			PRINTFORM 对于股布被捲起、内部笼罩着的%TEXTS("气味")%的湿气正在被观察的事完全不知情、
		ELSEIF RAND:2
			PRINTFORM 对于股布正在被脸摩擦着、微微飘散出的%TEXTS("气味")%被嗅到的事完全不知情、
		ELSE
			PRINTFORM 隔着勒入股間的衣料%TEXTS("气味")%被嗅到、\@ (TALENT:男性 || (TALENT:扶她 && RAND:2) ) ? 阴茎 # 阴部 \@被玩弄得無意識扭动着身体的同时、
		ENDIF
		PRINTFORMW 依然沉睡在夢中……
	ENDIF
CASE "撫摸股間"
	;裙や下摆で隠れた股間をまさぐった際には感触で身に着けている物を類推
	IF EQUIP:裙 && CFLAG:掀起裙子 == 0 && EQUIP:外套 == 0 && COND("胖次おぼろげ確認") == 0
		PRINTFORM 从手感上来说、今天的%CALLNAME:TARGET%……
		IF EQUIP:胖次 == 0
			PRINTFORMW 竟然、没有穿胖次！
		ELSEIF CHECK_PANTIES("Ｖ露出") || CHECK_PANTIES("Ａ露出")
			PRINTFORMW 竟然、穿着%DETAIL_PANTIES(TARGET, "感触")%的样子！
		ELSEIF CHECK_PANTIES("前貼")
			PRINTFORMW 穿的是贴在股間上的C字裤。
		ELSEIF CHECK_PANTIES("兜襠布")
			PRINTFORMW 好像系着兜襠布。
		ELSE
			PRINTFORMW 穿的应该是%DETAIL_PANTIES(TARGET, "感触")%。
		ENDIF
		CALL SETFLAG_DAILY, "胖次おぼろげ確認"
	ELSE
		IF CONDS("直前行動") == STR:日常イベント名 && CFLAG:躁动难忍 && ABL:露出癖 >= 5 && RAND:2
			PRINTFORMW %CALLNAME:TARGET%%TEXTR("流露出迷乱的眼神/露出了沉迷于欲望的表情/露出了兴奋的表情")%忍受着%CALLNAME:MASTER%的恶作剧、不断吐出甜腻的喘息……
		ELSEIF CFLAG:躁动难忍 || (ABL:露出癖 + CFLAG:甜言蜜语 && RAND:2)
			PRINTFORMW %CALLNAME:TARGET%不知廉恥地張開大腿、挺起腰接受了%CALLNAME:MASTER%的愛撫……
		ELSE
			PRINTFORM %CALLNAME:TARGET%%TEXTR("装出一副似乎在思考別的什麼事情的表情/雖然擺出了一副毫不在意的表情/沒有做出任何抵抗")%、
			PRINTFORMW 似乎是在邀請%CALLNAME:MASTER%一樣輕輕地打開了大腿……
		ENDIF
	ENDIF
CASE "掀起裙子", "撩起裙子"
	CALL TEXT_PANTU_CHECK_SKIRT
	IF TALENT:男性 && CFLAG:躁动难忍 && CHECK_PANTIES("男物") == 0
		PRINTFORM %CALLNAME:TARGET%的%TEXTR(@"女性用内衣/%PANTIES(TARGET, "種類")%")%
		SELECTCASE CONDS("阴茎尺寸")
		CASE "短小"
			PRINTFORM 的里面微微地膨起
			SIF ABL:露出癖 + ABL:欲望 >= 8
				PRINTFORM 小腹的附近还渗透出些许%TEXTR("汁液/先走汁/雄汁")%
			PRINTFORM が
		CASEELSE
			PRINTFORM 可以从边缘窥视到%TEXTR("先端/亀頭/先走汁/非常興奮")%的%TEXTR("怒張/阴茎/欲望")%
		ENDSELECT
		PRINTFORMW 、%TEXTR("在诉说着身为雄性而无法抑制的情欲/充满活力地主张着身为雄性的事实")%…
	ENDIF
ENDSELECT

TEXT_DETAIL =
;コンディション解説
SELECTCASE STR:日常イベント名
CASE "检查阴部"
	IF COND("合意：检查阴部")
		CALL PRINT_CUTIN, "すじ", TARGET
		SIF TEXTS("Ｖコンディション", TARGET) != ""
			CALL PRINT_STRW, @"(%CALLNAME:TARGET%的Ｖ状态是_黄色_%TEXTS("Ｖコンディション", TARGET)%_)"
	ENDIF
CASE "检查菊穴"
	SIF COND("合意：检查阴部") && TEXTS("Ａコンディション", TARGET) != ""
		CALL PRINT_STRW, @"(%CALLNAME:TARGET%的Ａ状态为_黄色_%TEXTS("Ａコンディション", TARGET)%_)"

CASE "股間格子"

;检查内褲とかでコンディション出すのはやりすぎかも。項目少なめだから地の文描写の方が良さげ
CASE "检查内褲"
	IF EQUIP:胖次 && COND("合意：检查内褲") && CHECK_PANTIES("開洞") && COND("胖次詳細確認") == 0
		SIF CHECK_PANTIES("勒紧的") || CHECK_PANTIES("Micro") || CHECK_PANTIES("高開衩") || CHECK_PANTIES("Ｃ胖次")
			TEXT_DETAIL += "[鼠径部露出]"
		SIF CHECK_CLO("胖次ハミ毛")
			TEXT_DETAIL += "[ハミ毛]"
		IF CHECK_PANTIES("透明")
			SIF COND("陰毛") > 3
				TEXT_DETAIL += "[透け毛]"
			SIF TALENT:陰蒂環 && PENIS(TARGET) == 0
				TEXT_DETAIL += "[透け阴蒂ピ]"
		ENDIF
		IF CHECK_STAIN(STAIN:胖次, "湿的")
			TEXT_DETAIL += "[愛液带污迹的]"
		ELSEIF CHECK_STAIN(STAIN:胖次, "闷热的", "汗")
			TEXT_DETAIL += "[ムレムレ]"
		ENDIF
		SIF CHECK_PANTIES("開洞带污迹的")
			TEXT_DETAIL += "[手ごわい带污迹的汚れ]"

		;SIF TEXT_DETAIL != ""
		;	CALL PRINT_STRW, @"(%CALLNAME:TARGET%の%PANTIES(TARGET, "胖次")%の状態は_黄色_%TEXT_DETAIL%_です)"
	ENDIF
;ここにはCHECK_OUTER("布鲁马")の場合に来る
CASE "检查布鲁马"
	IF CFLAG:股間確認済 == 0

	ENDIF
;休闲褲の時はあんまり？
CASE "股間観察"
	IF CFLAG:股間確認済 == 0

	ENDIF
;CASE "打開拉链"
;	CALL TEXT_PANTU_CHECK_SHORT
ENDSELECT

;胖次交易発生
SELECTCASE STR:日常イベント名
CASE "掀起裙子", "撩起裙子", "检查内褲", "股間観察"
	SIF CFLAG:睡眠 || CFLAG:オート胖次交易発生済 || EQUIP:胖次 == 0 || CHECK_CLO("胖次露出") == 0 || COND("合意：胖次交易") == 0
		RETURN 1

	CFLAG:オート胖次交易発生済 = 1
	CALL BLANKLINE
	PRINTFORML ……\@ CHARA_NUM("室内の邪魔者") + PLACE("野外") ? 断られるかもしれないが、 # \@今穿いている%PANTIES(TARGET, "胖次")%が欲しいと言ってみますか？
	CALL PRINT_SELECT, "言う/还是算了"
	SIF RESULT
		RETURN 1

	CALL EVENT_WISH_PANTU
ENDSELECT

@TEXT_PANTU_CHECK_SKIRT
CFLAG:掀起裙子 = 1
;胖次の上に何かを履いている場合
IF CHECK_CLO("连褲襪露出")
	CALL TEXT_PANTU_CHECK, SKIRT(TARGET), "连褲襪露出"
ELSEIF CHECK_CLO("外套露出")
	CALL TEXT_PANTU_CHECK, SKIRT(TARGET), "外套露出"
ELSEIF CHECK_CLO("贴身制服露出")
	CALL TEXT_PANTU_CHECK, SKIRT(TARGET), "贴身制服露出"
ELSEIF SKIRT(TARGET) != ""
	CALL TEXT_PANTU_CHECK, SKIRT(TARGET)
ELSE
	CALL TEXT_PANTU_CHECK_SHORT
ENDIF
CALL SETFLAG_DAILY, "掀起裙子"


code:
;-------------------------------------------------
;写真撮影
;製作者:hexagon
;写真の内容はSTR:写真とSTR:補足に格納し、実際には二つをくっつけた文字列が記録される
;例)		STR:写真 = 画面いっぱいに特写镜頭された比基尼胖次越しの濡烏の君の股間
;			STR:補足 = _特写镜頭_泳装
; 実際に保存される文字列 = 画面いっぱいに特写镜頭された比基尼胖次越しの濡烏の君の股間_特写镜頭_泳装
;追記:ブリート
;一部写真にデータを記録して画像を出力するようにしました
;そのため、STR:写真とSTR:補足の間に画像用の文字列が一部追加されます
;処理の関係により"STR:写真"と"STR:補足"内での"/","_","#"の文字の使用は注意してください
;-------------------------------------------------
@MESSAGE_DAILY_SELECT5
#DIM LCOUNT
#DIM EMPTY_MEMORY
#DIM NUM_MEMORY
#DIM MEMO_LINECOUNT

VARSET LOCALS
VARSET EMPTY_MEMORY
VARSET NUM_MEMORY

;メモリに余裕がない場合にはアルバム整理画面へ
IF NUM("写真") >= 300
	PRINTFORML 尽管还想再拍、但是已经存不下了。
	PRINTFORMW 请删掉不需要的写真閲覧画面。
	CALL VIEW_PHOTO
	RETURN 0
ENDIF

MEMO_LINECOUNT = LINECOUNT
STR:日常イベント名 = %DAILYCOMNAME(5)%

;起きている場合は変化することがある
IF CFLAG:睡眠 == 0
	SELECTCASE DAILYCOMNAME(5)
	CASE "已没收的胖次撮影"

	CASE "裙下撮影"

	CASE "股間撮影"

	CASE "内褲摄影"

	CASE "阴部摄影"

	CASE "肛門摄影"

	CASEELSE
		IF DEVENT("初回台詞") == 0
		ELSEIF COND("出浴") && CHOSE_DEVENT(STR:日常イベント名) == 0
		;たまに気づいてリアクションをとる
		ELSEIF COND("隐秘摄影可能") && (CFLAG:刚睡醒 || RAND:5 || CHOSE_DEVENT(STR:日常イベント名) == 0)
			IF RAND:10 == 0 || RAND:(CHOSE_DEVENT(STR:日常イベント名) + 1)
				STR:日常イベント名 = 隐秘摄影失敗
			ELSEIF RAND:2 && CFLAG:行動参加中 && CONDS("直前行動") != ""
				STR:日常イベント名 = 隐秘摄影失敗
			ENDIF
		ELSEIF COND("合意：摆POSE") && CFLAG:POSE结束摄影 == 0 && CFLAG:喧嘩中 == 0
			STR:日常イベント名 = POSE撮影
		ENDIF
	ENDSELECT
ENDIF

;-------------------------------------------------
;ここにはセリフのみを置く
;-------------------------------------------------
@MESSAGE_DAILY5
SIF TFLAG:地の文スキップ
	RETURN 0
STR:写真 =
STR:補足 =
SAVESTR:写真 =

SELECTCASE STR:日常イベント名
CASE "已没收的胖次撮影"
	PRINTFORMW %CALLNAME:MASTER%把从%CALLNAME:TARGET%没收得到的%PANTIES(TARGET, "胖次")%放在%NAMES("床")%上铺開。
	SWAP CSTR:胖次, CSTR:没収済胖次
	IF CHECK_PANTIES("前貼") || CHECK_PANTIES("Ｃ胖次")
		STR:写真 += @"貼在%CALLNAME:TARGET%身上的%PANTIES(TARGET)%"
	ELSE
		STR:写真 += @"%CALLNAME:TARGET%穿着の%PANTIES(TARGET)%"
	ENDIF
	SWAP CSTR:胖次, CSTR:没収済胖次

	STR:補足 += "_胖次"
CASE "裙下撮影"
	SELECTCASE CONDS("直前行動")
	CASE "掀起裙子"
		PRINTFORMW %CALLNAME:MASTER%把%CALLNAME:TARGET%的%CLO("掀起裙子")%掀起、拿出手机拍照。
		STR:写真 += @"%SKIRT(TARGET, "略")%被掀起、"
	CASE "撩起裙子"
		PRINTFORMW 在%CALLNAME:MASTER%的指示下、%CALLNAME:TARGET%自己用手撩起%CLO("掀起裙子")%准备拍摄照片。
		STR:写真 += @"%SKIRT(TARGET, "略")%被撩起、"
	ENDSELECT

	IF CHECK_CLO("胖次視認可能") == 0
		IF CHECK_CLO("贴身制服露出")
			STR:写真 += @"%CLO("股間")%紧密贴合着の股間"
		ELSE
			STR:写真 += @"%CLO("股間")%"
		ENDIF
		STR:補足 = _%SAVE_GRAPHICSDATE( CUTIN("見たまま正面", TARGET, 0) )%%STR:補足%
	ELSE
		;连褲襪描写
		SIF CHECK_CLO("连褲襪露出")
			STR:写真 += @"隔著%OUTER(TARGET)%露出"
		IF EQUIP:胖次 == 0
			IF COND("身嗜み：Ｖ拡張")
				STR:写真 += "拡張用振動棒挿入的"
			ELSEIF CHECK_CLO("Ｖ露出")
				STR:写真 += "赤裸的"
			ENDIF
			IF PENIS(TARGET)
				STR:写真 += "阴茎"
			ELSE
				STR:写真 += "阴部"
			ENDIF
		ELSEIF CHECK_PANTIES("Ｔ字褲")
			STR:写真 += @"勒进了%PANTIES(TARGET)%的\@ COND("巨尻") ? 大 # \@尻"
			STR:補足 += "_胖次"
			STR:補足 = _%SAVE_GRAPHICSDATE( CUTIN("見たまま正面", TARGET, 0) )%%STR:補足%
		ELSE
			STR:写真 += @"%PANTIES(TARGET)%"
			STR:補足 += "_胖次"
			STR:補足 = _%SAVE_GRAPHICSDATE( CUTIN("見たまま正面", TARGET, 0) )%%STR:補足%
		ENDIF
	ENDIF
	STR:写真 += @"\@ CONDS("直前行動") == "掀起裙子" ? 暴露 # 被展示 \@出来の%PHOTO_COSTUME(TARGET)%%CALLNAME:TARGET%"

CASE "股間撮影"
	PRINTFORM %CALLNAME:MASTER%
	IF CFLAG:睡眠
		PRINTFORMW 拿出手机靠近無防備地沉眠著の%CALLNAME:TARGET%の股間。
	ELSE
		PRINTFORMW 轻轻地把%CALLNAME:TARGET%的脚打開、拿出手机对准股间。
	ENDIF
	IF CHECK_CLO("连褲襪露出")
		IF EQUIP:胖次 && CHECK_PANTIES("開洞")
			STR:写真 += @"透过%OUTER(TARGET)%露出%PANTIES(TARGET)%の%CALLNAME:TARGET%的特写镜頭"
			STR:補足 += "_胖次_特写镜頭"
			STR:補足 = _%SAVE_GRAPHICSDATE( CUTIN("見たまま正面", TARGET, 0) )%%STR:補足%
		ELSE
			STR:写真 += @"隔著%OUTER(TARGET)%露出%CALLNAME:TARGET%の"
			IF PENIS(TARGET)
				STR:写真 += "阴茎"
				STR:補足 += "_阴茎"
			ELSE
				STR:写真 += "阴部"
				STR:補足 += "_阴部"
			ENDIF
		ENDIF
	;サービス
	ELSEIF COND("阴部のチラ見せ撮影可能") && CHECK_CLO("めくり可能") && (CHOSE_DEVENT(STR:日常イベント名) || RAND:3 == 0)
		STR:写真 += @"错開视线、主动翻開%CLO("股間")%展示阴部の%CALLNAME:TARGET%"
		STR:補足 += "_阴部"
		STR:日常イベント名 = 阴部のチラ見せ撮影
	ELSE
		STR:写真 += @"充满画面的特写镜頭：%CLO("股間")%"
		IF CLO("股間") != ""
			IF CHECK_CLO("勒紧的")
				STR:写真 += @"%CALLNAME:TARGET%深陷の股間"
			ELSE
				IF PENIS(TARGET)
					STR:写真 += @"穿着の%CALLNAME:TARGET%坚硬の股間"
				ELSE
					STR:写真 += @"穿着の%CALLNAME:TARGET%の股間"
				ENDIF
			ENDIF
		ELSE
			STR:写真 += @"%CALLNAME:TARGET%の%TEXTS("割れ目")%"
		ENDIF
		STR:補足 += "_特写镜頭"
		STR:補足 = _%SAVE_GRAPHICSDATE( CUTIN("見たまま正面", TARGET, 0) )%%STR:補足%
	ENDIF

CASE "内褲摄影"
	PRINTFORM %CALLNAME:MASTER%
	IF CFLAG:睡眠
		PRINTFORMW 拿出手机靠近無防備地沉眠著の%CALLNAME:TARGET%の股間。
	ELSE
		PRINTFORMW 以像是把脸埋进%CALLNAME:TARGET%股間的姿势取出了手机。
	ENDIF
	CALL PHOTO_PANTIES

CASE "肛門摄影"
	PRINTFORM %CALLNAME:MASTER%
	IF CFLAG:睡眠
		PRINTFORMW 拿出手机靠近無防備沉眠著%CONDS("間柄")%的、裸露的菊門。
	ELSE
		PRINTFORMW 拿出手机靠近暴露在眼前的、%CALLNAME:TARGET%の裸露的菊門。
	ENDIF
	CALL PHOTO_ANAL

CASE "阴部摄影"
	PRINTFORM %CALLNAME:MASTER%
	IF CFLAG:睡眠
		PRINTFORMW 拿出手机靠近無防備沉眠著%CONDS("間柄")%的、裸露的股間。
	ELSE
		PRINTFORMW 拿出手机对准在眼前暴露出的、%CALLNAME:TARGET%的裸露的股間。
	ENDIF
	CALL PHOTO_PUSSY

CASE "隐秘摄影失敗"
	PRINTFORMW %CALLNAME:MASTER%拿出手机、准备偷拍%CALLNAME:TARGET%。
	PRINTFORMW ……在撮影之前被%CALLNAME:TARGET%发现了！
	IF TALENT:叛逆
		STR:写真 += "神情愤怒の"
	ELSEIF TALENT:害羞 || TALENT:内気
		STR:写真 += "用手捂住脸の"
	ELSEIF TALENT:開朗 || TALENT:喜歡引人注目
		STR:写真 += "正在微笑の"
	ELSEIF TALENT:感情缺乏
		STR:写真 += "呆然若失の"
	ELSE
		STR:写真 += @"%TEXTR("呆呆笑着/面无表情")%の"
	ENDIF
	;服装と名前
	STR:写真 += @"%PHOTO_COSTUME(TARGET)%%CALLNAME:TARGET%"
CASE "POSE撮影"
	CALL PHOTO_POSING
CASEELSE
	;检查内褲系以外の寝てる時は寝顔を頂きます
	IF CFLAG:睡眠
		PRINTFORMW %CALLNAME:MASTER%用手机对准無防備沉眠著の%CALLNAME:TARGET%。
		IF PREVEX("口内射精")
			STR:写真 += @"从嘴角流下精液沉眠著の%CALLNAME:TARGET%"
		ELSEIF EXP_GAIN("睡眠姦経験")
			STR:写真 += @"%TEXTR("脸颊微微泛红/无意识地握着毛毯边缘/衣服被弄乱")%沉眠著の%CALLNAME:TARGET%"
		ELSEIF CONDS("直前行動") == "睡眠姦顔射"
			STR:写真 += @"被你的精液%TEXTR("变成白色/黏糊糊付着")%の%CALLNAME:TARGET%の寝顔"
		ELSE
			IF COND("快要醒了")
				STR:写真 += "稍微皱眉の"
			ELSE
				STR:写真 += @"%TEXTR("安静の/看起来很幸福の/可愛の")%"
			ENDIF
			STR:写真 += @"%CALLNAME:TARGET%の寝顔"
		ENDIF
	;ここから普通の写真
	ELSE
		PRINTFORMW %CALLNAME:MASTER%拿出手机、准备偷拍%CALLNAME:TARGET%。
		;最初の打招呼の時
		IF DEVENT("初回台詞") == 0
			STR:写真 += "正在作自我介绍の"
		;照れてる
		ELSEIF CONDS("直前行動") == "甜言蜜语"
			IF TALENT:恋慕
				STR:写真 += @"%TEXTR("兴奋得脸红/烧红了脸颊/着迷")%の"
			ELSE
				STR:写真 += @"%TEXTR("有些动摇/害羞地低着頭/看起来有些不满")%の"
			ENDIF
		;色々買ってきた
		ELSEIF CONDS("直前行動") == "買東西"
			STR:写真 += "将購入的"
			IF STRCOUNTS(CSTR:SHOP_BUY履歴, "避孕套") && (TALENT:好色 || TALENT:淫乱) && RANDIF(STRCOUNTS(CSTR:SHOP_BUY履歴, "振動棒"), 1)
				STR:写真 += "避孕套口叼在嘴里盯着这边の"
			ELSEIF STRCOUNTS(CSTR:SHOP_BUY履歴, "振動棒")
				IF ABL:受虐属性 >= 3 && STRCOUNTS(CSTR:SHOP_BUY履歴, "極太", "腕型", "馬型") && RAND:2
					STR:写真 +=  "凶悪大小的震动棒凝视着脸頬上泛着紅潮の"
				ELSEIF (TALENT:好色 || TALENT:淫乱) && RAND:2
					STR:写真 += "震动棒用舌頭缠住摆出挑衅动作の"
				ELSE
					STR:写真 += @"震动棒%TEXTR("用手指抚摸着/興味深地盯着/拿起来観察")%の"
				ENDIF
			ELSE
				STR:写真 += "商品饶有興味地打量着の"
			ENDIF
		;眠い
		ELSEIF CFLAG:刚睡醒
			IF TALENT:粗野 && RAND:3 == 0
				STR:写真 += "一边伸懒腰一边打哈欠の"
			ELSEIF TALENT:高贵 && RAND:3 == 0
				STR:写真 += "用手捂着嘴打哈欠の"
			ELSE
				STR:写真 += @"%TEXTR("刚起床就在发呆/还很困/无精打采")%の"
			ENDIF
		;暑い！
		ELSEIF COND("易出汗") && (COND("透け胸罩撮影可能") || CFLAG:PREV_DAILYCATE == 6 || RAND:3 == 0)
			IF COND("胸罩透け") && COND("透け胸罩撮影可能")
				;性処理直後
				SIF CFLAG:PREV_DAILYCATE == 6
					STR:写真 += @"%TEXTR("刚流下来的/完成工作后渗出来的")%の"
				IF EQUIP:胸罩 && CHECK_BRA("乳頭露出") == 0
					STR:写真 += @"汗水浸湿能看到%BRA(TARGET)%の"
				ELSE
					;乳頭が露出している胸罩
					IF EQUIP:胸罩
						STR:写真 += @"汗湿的%BRA(TARGET)%和"
					ELSE
						STR:写真 += "汗湿的"
					ENDIF
					IF TALENT:乳环 && RAND:2
						STR:写真 += @"%TEXTR("穿環反光的/穿環装饰着的")%乳頭"
					ELSEIF COND("巨乳輪") && RAND:2
						STR:写真 += @"%TEXTR("立派的/大号的")%乳輪"
					ELSEIF COND("乳頭肥大")
						STR:写真 += @"%TEXTR("大号/肥大的/立派的")%乳頭"
					ELSE
						STR:写真 += "乳頭"
					ENDIF
					STR:写真 += "隐约透露出来の"
				ENDIF
			ELSE
				;性処理直後
				SIF CFLAG:PREV_DAILYCATE == 6
					STR:写真 += @"%TEXTR("性処理完成/工作结束后/在努力的进行Ｈ的運動之后")%"
				IF COND("浑身是汗")
					STR:写真 += "全身是汗の"
				ELSE
					STR:写真 += "流了少量汗の"
				ENDIF
			ENDIF
		;淋浴でほかほか
		ELSEIF COND("出浴")
			STR:写真 += @"出浴后%TEXTR("頭发潮湿/脸上微微泛红/露出清爽表情")%の"
		;ポ萝莉
		ELSEIF CHECK_BRA("Micro") && EQUIP:服 == 0 && COND("巨乳輪") && RAND:4 == 0
			STR:写真 += "由于胸罩过小而露出大乳輪の"
		;超短迷你だと胖次チラも撮れる
		ELSEIF CHECK_SKIRT("超短迷你") && CHECK_CLO("胖次露出") && CONDS("直前行動") != "露脐装超短裙胖次チラ" && RAND:3 == 0
			STR:写真 += @"%SKIRT(TARGET, "略")%太短"
			IF EQUIP:胖次 == 0 || CHECK_PANTIES("前貼") || CHECK_PANTIES("Micro")
				STR:写真 += "屁股的一半"
				CALL SETFLAG_DAILY, "胖次おぼろげ確認"
			ELSE
				STR:写真 += @"%PANTIES(TARGET)%也"
				STR:補足 += "_胖次"
			ENDIF
			STR:写真 += "暴露出来の"
		;茶は機会多いので随机でスルー
		ELSEIF CFLAG:茶 && RAND:3 == 0
			IF RAND:2
				STR:写真 += @"%TEXTR("正在伸手拿/正在吃")%茶菓子の"
			ELSE
				STR:写真 += @"%TEXTR("一只手拿着茶杯/将茶杯放在一旁/将茶杯放到嘴边")%の"
			ENDIF
		;よっぱらい
		ELSEIF STATE("酔酒", TARGET) && RAND:2
			STR:写真 += @"醉得%TEXTR("眼睛发晕/脸颊泛红")%の"
		;做爱房とか情人旅館とか
		ELSEIF PLACE("出先")
			IF TALENT:好色 || TALENT:小悪魔 || TALENT:不知羞恥
				STR:写真 += "笑的很開心の"
			ELSEIF TALENT:感情缺乏 || TALENT:好奇心
				STR:写真 += "东张西望の"
			ELSE
				STR:写真 += "显得有些紧张の"
			ENDIF
		;おうちでゆっくり
		ELSEIF PLACE("自宅")
			SELECTCASE NO:TARGET
			;家事をする人たち。他にもいるかな？
			CASE 1, 20
				STR:写真 += @"%TEXTR("忙来忙去/干脆利落做着家务")%の"
			CASEELSE
				STR:写真 += @"%TEXTR("十分舒畅/不知为何看了过来/看上去有些懈怠")%の"
			ENDSELECT
		;部室でダラダラ
		ELSEIF PLACE("部室")
			STR:写真 += @"%TEXTR("在教室聊天/不知为何看了过来/正在打发时间")%の"
		ELSE

		ENDIF
		;服装と名前
		STR:写真 += @"%PHOTO_COSTUME(TARGET)%%CALLNAME:TARGET%"
	ENDIF
ENDSELECT

STR:補足 += @"_%CALLNAME:TARGET%"
SIF CHECK_CLO("泳装姿")
	STR:補足 += "_泳装"
SIF CHECK_CLO("体操服姿")
	STR:補足 += "_体操服"
SIF CHECK_CLO("学生服姿")
	STR:補足 += "_学生服"
SIF CFLAG:睡眠
	STR:補足 += "_睡眠中"

;日期の記録
STR:補足 += @"_日期%TOSTR(DAY*10 + TIME)%"

;起きている場合の撮影失敗判定
IF CFLAG:睡眠 == 0
	SELECTCASE DAILYCOMNAME(5)
	CASE "裙下撮影"
		IF CHECK_CLO("Ｖ露出")
			SIF COND("合意：Ｈ的写真") == 0
				STR:日常イベント名 = 撮影失敗
		ELSE
			SIF COND("合意：内衣写真") == 0
				STR:日常イベント名 = 撮影失敗
		ENDIF
	CASE "股間撮影"
		SIF COND("合意：内衣写真") == 0 && TALENT:不知羞恥 == 0
			STR:日常イベント名 = 撮影失敗
	CASE "内褲摄影"
		SIF COND("合意：内衣写真") == 0
			STR:日常イベント名 = 撮影失敗
	CASE "阴部摄影", "肛門摄影"
		SIF COND("合意：Ｈ的写真") == 0
			STR:日常イベント名 = 撮影失敗
	ENDSELECT
ENDIF

;撮影失敗ならここまで
SIF STR:日常イベント名 == "撮影失敗"
	RETURN 0

;補足次第でフラグ建て
SIF STRCOUNT(STR:補足, "胖次")
	CALL SETFLAG_DAILY, "胖次形状確認"

SAVESTR:PREV_PHOTO += @"\/%STR:日常イベント名%"

SAVESTR:写真 = %STR:写真%

SIF TFLAG:撮影おしまい == 0
	CALL SAVE_PHOTO, @"%STR:写真%%STR:補足%"

;相手に気づかれていない場合には時間経過が起きない
IF COND("気付いていない")
	PRINTFORML 咔嚓（攝像聲）
	PRINTFORML %CALLNAME:MASTER%は『%SAVESTR:写真%』を写真に収めた。

	;画像タグが埋め込んであったら画像を表示する
	SIF STRCOUNT(STR:補足, "GFX：")
		CALL LOAD_GRAPHICSDATE, EXTSTR(STR:補足, "GFX：", "_")

	WAIT
	;履歴には残す
	CSTR:DAILYCOM履歴 = %CSTR:DAILYCOM履歴%＠%STR:日常イベント選択名%\/
	STR:日常イベント名 =
ENDIF

;-------------------------------------------------
;写真撮影後に何かイベントを起こしたい場合はここに書いても良い
;-------------------------------------------------
@AFTERMESSAGE_DAILY5
SELECTCASE STR:日常イベント名
CASE "撮影失敗"
	PRINTFORM ……%CALLNAME:TARGET%似乎对
	SELECTCASE DAILYCOMNAME(5)
	CASE "裙下撮影"
		IF CHECK_SKIRT("裙")
			PRINTFORM 裙下
		ELSEIF CHECK_CLO("Ｖ露出")
			PRINTFORM 重要地方的
		ELSE
			PRINTFORM 这种状态的
		ENDIF
	CASE "股間撮影"
		SIF CHECK_CLO("外套露出") || CHECK_CLO("贴身制服露出") || EQUIP:胖次
			PRINTFORM 穿着%CLO("股間")%、
		PRINTFORM 股间的
	CASE "内褲摄影"
		PRINTFORM 内衣的
	CASE "肛門摄影"
		PRINTFORM 肛門的
	CASE "阴部摄影"
		PRINTFORM 重要地方的
	ENDSELECT
	PRINTFORMW 摄影有抵抗情绪。
	PRINTFORMW %CALLNAME:MASTER%的写真摄影失败了。
CASE "隐秘摄影失敗"
	PRINTFORML 咔嚓（攝像聲）
	PRINTFORMW %CALLNAME:MASTER%保存了『%SAVESTR:写真%』的照片！
	;PRINTFORMW ……この時間帯で%CALLNAME:TARGET%を隐秘摄影するのは、もう無理そうだ。
CASE "阴部のチラ見せ撮影"
	PRINTFORM 察觉到自己正在被摄影的%CALLNAME:TARGET%%PHOTO_SHOWING()%将手指伸向镜頭所指的位置、
	SELECTCASE DAILYCOMNAME(5)
	CASE "股間撮影"
		PRINTFORM 移開布料
	CASE "内褲摄影"
		PRINTFORM 翻開布料
	ENDSELECT
	PRINTFORMW 将%TEXTS("割れ目")%暴露在镜頭前。
	PRINTL
	PRINTFORML 咔嚓（攝像聲）
	PRINTFORMW %CALLNAME:MASTER%保存了『%SAVESTR:写真%』的照片！
CASE "肛門の見せつけ撮影"
	IF COND("身嗜み：Ａ拡張") && STRCOUNT(STR:補足, "肛門")
		PRINTFORM 察觉到自己的尻穴正在被摄影的%CALLNAME:TARGET%%PHOTO_SHOWING()%将塞住肛門的
		SELECTCASE COND("身嗜み：Ａ拡張")
		CASE IS >= 4
			PRINTFORM 特大号肛塞一口气
		CASE 3
			PRINTFORM 极粗的肛塞一口气
		CASEELSE
			PRINTFORM 肛門塞子一口气
		ENDSELECT
		PRINTFORMW 拔出、%TEXTR("松弛了的/鬆弛下來的/大開的")%肛門在%CALLNAME:MASTER%面前暴露出来。
	ELSE
		PRINTFORM 察觉到自己的尻穴正在被摄影的%CALLNAME:TARGET%%PHOTO_SHOWING()%把手伸向臀部、
		PRINTFORMW %TEXTR("一下子/快速地/刷的一下")%把肛門推到镜頭面前。
	ENDIF
	PRINTL
	PRINTFORML 咔嚓（攝像聲）
	PRINTFORMW %CALLNAME:MASTER%保存了『%SAVESTR:写真%』的照片！
CASE "阴部の見せつけ撮影"
	IF COND("身嗜み：Ｖ拡張") && STRCOUNT(STR:補足, "膣")
		PRINTFORM 察觉到自己的阴部正在被摄影的%CALLNAME:TARGET%%PHOTO_SHOWING()%将膣内埋藏的
		SELECTCASE COND("身嗜み：Ｖ拡張")
		CASE IS >= 4
			PRINTFORM %TEXTR("笨重的玩具/大约和手臂一样粗的振動棒/20cm以上的極太振動棒")%一口气
		CASE 3
			PRINTFORM %TEXTR("拡張用玩具/極太振動棒/略小于20cm的極太振動棒")%一口气
		CASE 2
			PRINTFORM %TEXTR("巨根模型玩具/大号的振動棒/15cm程度的大型振動棒")%一口气
		CASEELSE
			PRINTFORM %TEXTR("拡張用玩具/相当于男性器大小的振動棒/振動棒")%一口气
		ENDSELECT
		PRINTFORM 抜出、%TEXTR("累积的爱液/原先被阻挡住的愛液/高粘度的愛液")%%TEXTR("不斷垂落/大量溢出/全部漏出")%
		PRINTFORMW %TEXTR("松弛了的/鬆弛下來的/大開的")%膣口在%CALLNAME:MASTER%面前暴露出来。
	ELSE
		PRINTFORM 察觉到自己的股間正在被摄影的%CALLNAME:TARGET%%PHOTO_SHOWING()%将手指伸入%TEXTS("割れ目")%、
		PRINTFORMW %TEXTR("将外側的褶皱和/轻轻地将/一下子把")%内部的粘膜推開展示在镜頭前。
	ENDIF
	PRINTL
	PRINTFORML 咔嚓（攝像聲）
	PRINTFORMW %CALLNAME:MASTER%保存了『%SAVESTR:写真%』的照片！
CASE "POSE撮影"
	IF TFLAG:撮影おしまい
		PRINTFORMW %CALLNAME:TARGET%用夸张的調子摆出『结束』的手势……撮影很成功的样子。
		CFLAG:POSE结束摄影 = 1
	ELSE
		PRINTFORML 咔嚓（攝像聲）
		PRINTFORML %CALLNAME:MASTER%は『%SAVESTR:写真%』を写真に収めた。

		;画像タグが埋め込んであったら画像を表示する
		SIF STRCOUNT(STR:補足, "GFX：")
		CALL LOAD_GRAPHICSDATE, EXTSTR(STR:補足, "GFX：", "_")

		WAIT
	ENDIF
CASEELSE
	PRINTFORML 咔嚓（攝像聲）
	PRINTFORML %CALLNAME:MASTER%は『%SAVESTR:写真%』を写真に収めた。

	;画像タグが埋め込んであったら画像を表示する
	SIF STRCOUNT(STR:補足, "GFX：")
		CALL LOAD_GRAPHICSDATE, EXTSTR(STR:補足, "GFX：", "_")

	WAIT
ENDSELECT

license: unknown